This first code is the code of the transmitter (the card connected to the computer). 

'{$STAMP BS2}
' {$PBASIC 2.5}

'     ===================================================
'=====Wireless control of the BOEBOT. Transmitter program=====
'     ===================================================


'--------- Variables -----------

InVar              VAR Nib
UtVar              VAR Nib
PauVar             VAR Byte
Ini                VAR Nib


'--------- Pins -----------

Sandare            PIN 0        'Transmitter pin
Knapp1             PIN 1        'Button1
Summer             PIN 2        'Buzzer


'--------- Main Loop ------------

UtVar = 5
Ini = 0
PauVar = 20

FREQOUT Summer,250,1000

start:

DO
DEBUG CLS
DEBUG "Invar = ", DEC Invar, CR
DEBUG "Utvar = ", DEC Utvar, CR

SEROUT Sandare, 3313, [utVar]    'Transmit value from utVar to the BOEBOT
PAUSE PauVar

IF (Knapp1 = 1) THEN Ini = 1
IF Ini = 1 THEN program

LOOP

Program:

SERIN 16, $4054, 20, start, [InVar]         'Receive value from the computer, store in InVar
PAUSE PauVar
utvar = invar

GOTO start

When the program starts it sends the value "5" over and over again until you push "Knapp1" (button 1) on the card. This is used to synchronize the communication between the transmitter and the receiver. When you push the button the it starts to receive values that you enter in the debug window in BASIC Stamp Editor.

 

This is the robot program code.

'{$STAMP BS2}
'{$PBASIC 2.5}

'     =====================================
'=====Wireless control of the BOEBOT. Robot (receiver) program=====
'     =====================================


'----- Variables -----

InVar              VAR Nib
IniVar             VAR Nib

Ini                VAR Nib

MyVar              VAR Nib
InPutVar           VAR Nib

irDetectLeft       VAR Bit
irDetectRight      VAR Bit
pulsecount         VAR Byte


'--------- Pins -----------

Mottagare          PIN 1                    'Receiver
Summer             PIN 2                    'Buzzer

IrInLeft           PIN 10
IrInRight          PIN 3
IrLedLeft          PIN 11
IrLedRight         PIN 4

Right_Servo_Wheel  PIN 12
Left_Servo_Wheel   PIN 13

IniLED             PIN 9

'----- Constants -----

CRMid              CON 750
CRDeviation        CON 225
CRMax              CON CRMid + CRDeviation
CRMin              CON CRMid - CRDeviation
CRHMid             CON CRMid + 20
CRLMid             CON CRMid - 20

CLMid              CON 750
CLDeviation        CON 225
CLMax              CON CLMid + CLDeviation
CLMin              CON CLMid - CLDeviation
CLHMid             CON CLMid + 20
CLLMid             CON CLMid - 20

Stopped            CON 5
Forward            CON 8
back               CON 2
Turn_Right_Frwd    CON 9
Turn_Left_Frwd     CON 7
Turn_Right_Sharp   CON 6
Turn_Left_Sharp    CON 4
Turn_Right_Back    CON 3
Turn_Left_Back     CON 1
Ir_styrd           CON 12

Ini = 0
LOW 15
Inivar = 5
myvar = 5

FREQOUT Summer,250,1000

'----- Initiating -----
ini_loop:
DO
  DEBUG CLS                                         'Clears the debug-window
  DEBUG "Invar = ", DEC Invar, CR                   'Show the In-variable in the debug-window
  DEBUG "Ini = ", DEC Ini, CR
  DEBUG "Myvar = ", DEC myVar, CR

  SERIN Mottagare, 3313, [InVar]                    'Receive value and store it in the variable InVar
  PAUSE 20                                          'Pause 20 ms

  IF Invar <> Inivar AND Ini=0 THEN Reset           'Jumps to the Reset-loop if not the initiation-value is received
  IF Invar = Inivar AND Ini=0 THEN Ini = 1          'Makes "Ini" high if the initiation-value is received

  IF Ini = 1 THEN HIGH IniLED
  IF Ini = 1 THEN GOTO start                        'Jumps to the main program when the communication is initiaded

LOOP

'----- Reset loop -----
Reset:

SLEEP 1                                   '"Sleeps" for one timeunit, ca 2.3s
GOTO ini_loop

'----- Serial communication loop -----
start:
  DEBUG CLS                                         'Clears the debug-window
  DEBUG "Invar = ", DEC Invar, CR                   'Shows the In-variable in the deboug-window
  DEBUG "Ini = ", DEC Ini, CR
  DEBUG "Myvar = ", DEC myvar, CR

  SERIN Mottagare, 3313, [InVar]                    'Receive value and store it in the variable InVar
  LOW Iniled
  PAUSE 20                                          'Pause 20 ms
  HIGH iniled

  GOTO program

'----- Main program loop -----
Program:

myvar = InVar
SELECT myvar

   CASE = Stopped
      GOSUB still

   CASE = Forward
       PULSOUT Right_Servo_Wheel, CRMin
       PULSOUT Left_Servo_Wheel , CLMax

   CASE = Back
       PULSOUT Right_Servo_Wheel, CRMax
       PULSOUT Left_Servo_Wheel , CLMin

   CASE = Turn_Right_frwd
       PULSOUT Right_Servo_Wheel, CRLMid
       PULSOUT Left_Servo_Wheel , CLMax

   CASE = Turn_Left_frwd
       PULSOUT Right_Servo_Wheel, CRMin
       PULSOUT Left_Servo_Wheel , CLHMid

   CASE = Turn_Right_Sharp
       PULSOUT Right_Servo_Wheel, CRMax
       PULSOUT Left_Servo_Wheel , CLMax

   CASE = Turn_Left_Sharp
       PULSOUT Right_Servo_Wheel, CRMin
       PULSOUT Left_Servo_Wheel , CLMin

   CASE = Turn_Right_back
       PULSOUT Right_Servo_Wheel, CRHMid
       PULSOUT Left_Servo_Wheel , CLMin

   CASE = Turn_Left_back
       PULSOUT Right_Servo_Wheel, CRMax
       PULSOUT Left_Servo_Wheel , CLLMid

   CASE = Ir_styrd
      GOTO Ir

   CASE ELSE
      GOSUB still

ENDSELECT

GOTO start

'----- IR-controlled -----


Ir:

  FREQOUT IrLedLeft, 1, 38500
  irDetectLeft = IrInLeft

  FREQOUT IrLedRight, 1, 38500
  irDetectRight = IrInRight

  IF (irDetectLeft = 0) AND (irDetectRight = 0) THEN         'Both IR, back up, turn left
    GOSUB Back_Up
    GOSUB Turn_left
    GOSUB Turn_left
  ELSEIF (irDetectLeft = 0) THEN                   'Left IR, right turn
    GOSUB back_up
    GOSUB turn_right
  ELSEIF (irDetectRight = 0) THEN                   'Right IR, left turn
    GOSUB back_up
    GOSUB turn_left
  ELSE
    GOSUB Forward_pulse                   'No IR, forward
  ENDIF
GOTO Start


'----- Subroutine -----

Forward_Pulse:
  FOR pulsecount = 0 TO 10
   PULSOUT 13, 850
   PULSOUT 12, 650
   PAUSE 20
  NEXT
  RETURN

Turn_Left:
  FOR pulsecount = 0 TO 20
    PULSOUT 13, 650
    PULSOUT 12, 650
    PAUSE 20
  NEXT
  RETURN

Turn_Right:
  FOR pulseCount = 0 TO 20
    PULSOUT 13, 850
    PULSOUT 12, 850
    PAUSE 20
  NEXT
  RETURN

Back_Up:
  FOR pulseCount = 0 TO 40
    PULSOUT 13, 650
    PULSOUT 12, 850
    PAUSE 20
  NEXT
  RETURN

still:
    PULSOUT 13, 750
    PULSOUT 12, 750
  RETURN

First of all. Yes I have copyed alot from a program called "Tethered BOEBOT Control using SERIN from Debug", so if you who made that program reads this. Thank you very much, your program helped my alot.

Back to my program. When the program starts it tries to receive the value "5", if it doesn't receive it the program sleeps for about 2.3 seconds and then it starts over. When the program receive the value "5" the "IniLED" turns high and then you know that the transmitter and the receiver is synchronized and you can push the button1 on the transmitter board. Then the program jumps to different subroutins depending on what number you enter in the debug-window.

As I said, there is alot of work left before the program is finished so please, be not disappointed if your robot not always obey the commands that you give it. Maybe it is the transmitter/receiver that isn't the best but for that money I spent I think it works better than I hoped.

Please enjoy!!!

